Lecture 15

Advanced Screen Processing

Text:
5th Edition: Chapter 9
4th Edition: Chapter 10

VIDEO in the computer



Video Display Pages
(Default is page 0)

Each character on the screen (in text mode) has

- ASCII character value (1 byte)
- Attribute (1 byte)

The size of the video memory is 4096 (4K, 1000h) bytes:
25 rows $\times$ 80 columns $\times$ 2 bytes = 4000 bytes

Attribute bytes:

- foreground blinking (1 bit)
- background color RGB (3 bits)
- foreground intensity I (1 bit)
- foreground color RGB (3 bits)

Color:

	Intensity I=0		Intensity I=1
Color	R G B	Color	R G B
Black	0 0 0	Gray	0 0 0
Blue	0 0 1	Light Blue	0 0 1
Green	0 1 0	Light Green	0 1 0
Cyan	0 1 1	Light Cyan	0 1 1
Red	1 0 0	Light Red	1 0 0
Magenta	1 0 1	Light Magenta	1 0 1
Brown	1 1 0	Yellow	1 1 0
White	1 1 1	High Int White	1 1 1

The attribute byte:

			Background				Foreground		
Background	Foreground	Bl	R	G	B	I	R	G	B
Black		0	0	0	0				
	Black					0	0	0	0
Blue	Red	0	0	0	1	0	1	0	0
Green	Yellow	0	0	1	0	1	1	1	0
White	Gray blink	1	1	1	1	1	0	0	0
Magenta	Lt. Magenta	0	1	0	1	1	1	0	1
White	White blink	1	1	1	1	0	1	1	1

BIOS INT 10h for text mode
Function Description (AH)

00h Set Video Mode (AL=03h for 80X25 color text)
01h Set Cursor Size
02h Set Cursor Position
03h Read Cursor Position
05h Select Active Page
06h Scroll Up Screen
07h Scroll Down Screen
08h Read Attribute and Character at Cursor Position
09h Display Attribute and Character at Cursor Position
0Ah Display Character at Cursor Position
0Eh Write Teletype
0Fh Get Current Video Mode
13h Display character string

Example: function code 13h: Display string

```
STACKSG  SEGMENT  PARA  STACK 'Stack'
         DW       32
STACKSG  ENDS
DATASG   SEGMENT  PARA 'Data'
MyName   DB      'Calvin Coolidge'
DATASG   ENDS
CODESG   SEGMENT  PARA 'Code'
BEGIN    PROC     FAR
      ASSUME SS:STACKSG,DS:DATASG,CS:CODESG
      MOV   AX,DATASG
      MOV   DS,AX
      MOV   ES,AX
;
      MOV   AH,13H    ;DISPLAY CHAR STRING
      MOV   AL,01     ;DISP ATT& ADV CURSOR
      MOV   BH,00     ;PAGE 0
      MOV   BL,9Fh    ;BLNK INT WH ON BLUE
      LEA   BP,MyName ;STRING TO DISPLAY
      MOV   CX,15     ;LENGTH OF STRING
      MOV   DX,0D20h  ;at row 13  column 32
      INT   10H

; exit to DOS
       MOV      AX,4C00H
       INT      21H
BEGIN  ENDP
CODESG ENDS
       END      BEGIN
```

This program displays the character string “Calvin Coolidge” in blinking intense white on blue starting at row 25, column 32. (Attribute 9Fh = 10011111)

```
TITLE   P10BIOAS Use BIOS to disp ASCII char set
        .MODEL SMALL    .COM
        .CODE
        ORG     100H
BEGIN:  JMP     SHORT MAIN
CTR     DB      00      ;Counter for ASCII chars
COL     DB      24      ;Column of screen
ROW     DB      04      ;Row of screen
MODE    DB      ?       ;Video mode
;       Main procedure:
;       --------------
MAIN    PROC    NEAR
        CALL    B10MODE ;Get/set video mode
        CALL    C10CLR  ;Clear screen
A20:
        CALL    D10SET  ;Set cursor
        CALL    E10DISP ;Display chars
        CMP     CTR,0FFH ;Last char displayed?
        JE      A30     ;  yes -- exit
        INC     CTR     ;Increment ASCII counter
        ADD     COL,02  ;Increment column
        CMP     COL,56  ;At end of column?
        JNE     A20     ;  no -- bypass
        INC     ROW     ;  yes - increment row
        MOV     COL,24  ;  and reset column
        JMP     A20
A30:
        CALL    F10READ ;Get keyboard char
        CALL    G10MODE ;Restore video mode
        MOV     AX,4C00H ;Exit to DOS
        INT     21H
MAIN    ENDP
```

```
;       Get and set video mode
;       ----------------------
B10MODE PROC   NEAR
        MOV     AH,0FH  ;Request get mode
        INT     10H
        MOV     MODE,AL ;Save mode
        MOV     AH,00H  ;Request set new mode
        MOV     AL,03   ;Standard color
        INT     10H
        RET
B10MODE ENDP
```

```
;       Clear screen and create window:
;       -------------------------------
C10CLR  PROC   NEAR
        MOV     AH,08H  ;Request get current
        INT     10H     ;  attribute in AH
        MOV     BH,AH   ;Move it to BH
        MOV     AX,0600H ;Scroll whole screen
        MOV     CX,0000 ;Upper left location
        MOV     DX,184FH ;Lower right location
        INT     10H
        MOV     AX,0610H ;Create 16-line window
        MOV     BH,16H   ;Brown on blue
        MOV     CX,0418H ;Upper left corner 04:24
        MOV     DX,1336H ;Lower rt cornr 19:54
        INT     10H
        RET
C10CLR  ENDP
```

```
;       Set cursor to row and column:
;       ----------------------------
D10SET  PROC   NEAR
        MOV     AH,02H  ;Request set cursor
        MOV     BH,00   ;Page 0 (normal)
        MOV     DH,ROW  ;New row
        MOV     DL,COL  ;New column
        INT     10H
        RET
D10SET  ENDP
```

```
;       Display ASCII chars:
;       -----------------------
E10DISP PROC   NEAR
        MOV     AH,0AH  ;Display
        MOV     AL,CTR  ;ASCII char
        MOV     BH,00   ;Page 0
        MOV     CX,01   ;One char
        INT     10H
        RET
E10DISP ENDP
```

```
;       Force pause, get keyboard char
;       ----------------------------------
F10READ PROC   NEAR
        MOV     AH,10H  ;Request get char
        INT     16H
        RET
F10READ ENDP
```

```
;       Restore original video mode
;       ---------------------------
G10MODE PROC   NEAR
        MOV     AH,00H  ;Request set mode
        MOV     AL,MODE ;Original value
        INT     10H
        RET
G10MODE ENDP
        END     BEGIN
```

EXTENDED ASCII CHARACTERS

insert table B-1, Page 546

A program to draw a box with text in it:

```
STACKSG SEGMENT  PARA  STACK 'Stack'
         DW       32
STACKSG  ENDS
DATASG  SEGMENT  PARA 'Data'
CR       EQU     0DH
LF       EQU     0AH
BOX     DB  0DAh,18 DUP(0C4h),0BFh,CR,LF
        DB  0B3h,'    Welcome to    ',0B3h,CR,LF
        DB  0B3h,' Computer Science ',0B3h,CR,LF
        DB  0B3h,'   at   ',0B3h,CR,LF
        DB  0B3h,'  Queens College  ',0B3h,CR,LF
        DB  0C0h,18 DUP(0C4h),0D9h,CR,LF

DATASG   ENDS
CODESG   SEGMENT  PARA 'Code'
BEGIN    PROC     FAR
         ASSUME   SS:STACKSG,DS:DATASG,CS:CODESG
         MOV      AX,DATASG
         MOV      DS,AX
         MOV      ES,AX
;
         MOV      AH,40H
         MOV      BX,01  ;FILE HANDLE FOR SCREEN
         MOV      CX,132 ;TOTAL CHARACTERS 6*22
         LEA      DX,BOX
         INT      21H

; exit to DOS
         MOV      AX,4C00H
         INT      21H
BEGIN    ENDP
CODESG   ENDS
         END      BEGIN
```

Exercises - Lecture 15

1. Write a program which prompts the user to enter a character that is either "R", "G", or "B". After the user enters one of these characters, clear the screen with that background color.

2. Write a program that draws a box in the center of the screen which is 15 columns wide and 10 rows high. The inside of the box should be yellow, and the rest of the screen should be green.